A MICRO-COMPUTER-BASED SYSTEM
TO COMPUTE MAGNETIC VARIATION*

Rajan Kaul
Avionics Engineering Center
Department of Electrical and Computer Engineering
Ohio University
Athens, Ohio

*Published as Ohio University Technical Memorandum 91, March 1984.

A micro-computer-based implementation of a magnetic variation model for the continental United States is presented. The implementation computes magnetic variation as a function of latitude and longitude for general aviation receivers such as Loran-C.

I. INTRODUCTION

A mathematical model of magnetic variation in the continental United States (COT48) has been implemented in the Ohio University Loran-C receiver. The model is based on a least squares fit of a polynomial function. The implementation on the micro-processor based Loran-C receiver is possible with the help of an Am 9511 math chip, manufactured by Advanced Micro Devices, which performs 32 bit floating point mathematical operations. A Peripheral Interface Adapter (M6520) is used to communicate between the 6502 based micro-computer and the 9511 math chip. The implementation provides magnetic variation data to the pilot as a function of latitude and longitude. This report briefly describes the model and the real time implementation in the receiver.

II. THE MATHEMATICAL MODEL

The model was developed at the United States Geological Survey (USGS) by Fabiano et al. [1], by performing least squares analysis on more than 34,000 data measurements taken between 1900 and 1974. The analysis provides an analytical model of the magnetic field which is used to compute the magnetic variation.

For the actual magnetic variation calculation the COT48 region is partitioned into five, 12 degree longitudinal bands from 66 degrees West to 126 degrees West. A set of coefficients for each of the five bands is determined by the analytical model. A seventh order polynomial function of the analytical model is applied to compute the magnetic variation. The secular change is calculated in a similar way, the only difference being that a sixth order polynomial function is used. Also, the secular change case is not partitioned into bands and therefore the same set of coefficients is used for the entire COT48 region.

The polynomial function adapted for the procedure is

$$\sum_{i=0}^{n} \sum_{j=0}^{i} a_{ij} (\Theta_c)^{i-j}(\lambda_c)^j$$

a_{ij} - co-efficients

Θ_c - normalized latitude $= \Theta - 52$

λ_c - normalized longitude $\lambda - \gamma$

γ - Table 1 - east longitude normalizing factor

Θ - co-latitude = 90°-latitude

λ - east longitude = 360°-longitude

The limits on each band and other constants are specified in Table 1 [2].

For the magnetic variation calculation n=7, and thus 36 coefficients are specified for each band in the COT48 region, while in the secular change calculation n=6, therefore only 28 coefficients are required for the whole COT48 region. All the coefficients are given in Appendix A.

The model was simulated in FORTRAN on an IBM 370 computer at Ohio University and a contour plot was made of the magnetic variation in the COT48 region (figure 1). The FORTRAN program listing is included in Appendix B. A copy of an actual magnetic variation chart published by the Defense Mapping Agency (DMA) is shown in figure 2. Comparisons between actual published values of the magnetic variation and values calculated by the model were made and are described later in this report.

III. MICRO-COMPUTER IMPLEMENTATION

The magnetic variation model was implemented on a 6502 based Super-Jolt micro-computer. The 6502 microprocessor has only an 8-bit data bus, so the processor needs a large amount of memory and rapid access. The calculations in the implementation of the model require complex floating point operations of exponents. It is therefore desirable to use an external hardware device to support the microprocessor in these calculations.

The Am9511 was chosen to be implemented with the Super-Jolt system. It is a peripheral math processor which performs the necessary floating point mathematical operations. The Am9511 is designed to be used in conjunction with microprocessor systems that have an 8-bit data bus. The stack oriented processor can handle 16 and 32-bit floating point operands and performs arithmetic and trigonometric functions. An instruction set of the Am9511 is included in Appendix C [3].

Additional hardware is necessary to allow the microprocessor to communicate with the math processor. An M6520 peripheral Interface Adapter (PIA) is used for handshaking with the microcomputer. The PIA consists of two 8-bit ports and several control registers for interface with external support devices. The overall design of the microcomputer, which is a part of the Ohio University Loran-C project is shown in figure 3.

IV. INTERFACING SOFTWARE

Special software is needed to allow the hardware components to interact with one another. The four subroutines 'PINT', 'PUSH', 'POP' and 'CMND' were written by Fischer [4] with this particular goal in mind. 'PINT' initializes the Am9511 and the PIA and, also, the scratchpad RAM locations. 'PUSH' is used to copy a four byte number from RAM to the stack of the Am9511. 'POP' does exactly the opposite by copying a four byte floating point number from the stack of the Am9511 to scratchpad RAM. 'CMND' sends an instruction byte to the Am9511 to perform the desired operation. It also checks the status register of the math processor to determine the outcome of the operation. Flow charts of the above subroutines are given in figure 4.

The actual magnetic variation program 'MAGVAR', occupies about 800 bytes of memory including scratchpad RAM locations. The coefficients 36 for each of the five bands, occupy 900 bytes of memory. Each of the coefficients is converted into a 32-bit floating point format compatible with the Am9511 representing four bytes. The secular variation calculation is not included in the real time implementation for reasons which shall be addressed later in this report. The complete program listing is given in Appendix D at the end of this report. The 'MAGVAR' program takes about 1.5 seconds in execution time. However, since the magnetic variation does not change rapidly in a small geographic region, it does not need to be computed every time navigation position information is updated, when included in the actual navigation receiver such as the Ohio University Loran-C. For example a small change in the software can allow computation of the magnetic variation every 30 miles, or a one degree change in geographic position or any other interval desired.

V. RESULTS AND CONCLUSIONS

Initially, the values for the magnetic variation were computed by the FORTRAN simulation and compared to values published by National Geophysical Data Center.* The results obtained were accurate to a large degree. Table 2 summarizes two points in each band, of which comparisons were made in the COT48 region. The reason for the discrepancy in the values could arise from the differences between the data and the model. Fabiano and others [1] evaluated the model and compared it to surveyed data for 1,450 points. From these measurements an overall root mean squared deviation of 0.5 degrees was found in the magnetic variation in the COT48 region. Also, a probable cause for the larger discrepancy in the region of bands 2 and 3 could indicate magnetic variation anomalies in the Great Lakes region.

In general, the results were found to be satisfactory and the decision was made to implement the model on the Ohio University Loran-C receiver. The results computed by the microcomputer were within 0.1 degrees of the values computed by the FORTRAN simulation. As indicated earlier, the secular change was not implemented on the receiver. The magnetic variation in the COT48 region changes less than 11 minutes of arc annually at its worst case. This translates to a change of less than one degree over a period of five years at its worst case. Since the Ohio University Loran-C receiver is a research tool, not implementing the secular change function would not have a crucial impact on the outcome of future research. The coefficients for the model are derived every five years by the USGS, and can be updated very easily to keep the model current.

The overall performance of the implementation proves to be satisfactory. The major advantages are that the magnetic variation is available all the time to the pilot to allow accurate determination of the compass heading. It is computed automatically and is one less adjustment or source of error during a flight, thus also reducing the chances of pilot error.

VI. ACKNOWLEDGMENTS

The work presented in this technical memorandum has been supported by the National Aeronautics and Space Administration at Langley Research Center under grant number NGR 36-009-017. It was performed at Ohio University's Avionics Engineering Center.

*National Geophysical Data Center, Boulder, Colorado.

The author would like to acknowledge the help of Dr. Robert Lilley, Associate Director and Mr. James Nickum, project engineer at Avionics Engineering center, whose suggestions proved invaluable in all stages of this research. Special thanks is also due to Dr. Hugh Bloemer, Department of Geography, whose advice was indispensable.

VII. REFERENCES

[1] Fabiano, E. B., W. J. Jones, N. W. Peddie, "The Magnetic Charts of the United States for Epoch, 1975, "United States Geological Survey, Circular No. 810.

[2] Ibid., Fabiano, Jones and Peddie.

[3] "Am9511A Arithmetic Processor Advanced Micro Devices Advanced MOS/LSI," Advanced Micro Devices, Inc., Sunnyvale, California, 1976.

[4] Fischer J.P., "A Microcomputer-based Position Updating System for General Aviation Utilizing Loran-C," M.S. Thesis, Ohio University, Athens, Ohio, March 1982.

VIII. APPENDICES

A. Co-efficients for the 5-band and secular change in COT48.

B. FORTRAN Program Listing of "MAGVAR".

C. Instruction Set for the Am9511.

D. 6502 Assembly Language Program Listing of the Magnetic Variation Implementation on the Ohio University Loran-C Receiver.

Table 1. Limits on Five Bands and the East Longitude Normalizing Factor.

Band	Partition °W longitude	λ(degrees)
1	66-77	289
2	78-89	277
3	90-101	265
4	102-113	253
5	114-125	241

Table 2. Comparisons Between Actual and Computed Values of Magnetic Variation.

Band	Latitude Deg. N	Longitude Deg. W	Magnetic Variation Actual	Magnetic Variation Computed
1	36	77	7.72°W	7.40°W
66°W - 78°W	40	73	13.05°W	12.71°W
2	32	81	3.98°W	2.78°W
78°W - 90°W	40	87	1.12°W	0.51°W
3	34	93	4.37°E	5.90°E
90°W - 102°W	38	101	9.05°E	10.84°E
4	36	103	9.93°E	10.91°E
102°W - 114°W	40	107	12.62°E	13.54°E
5	38	123	16.18°E	16.59°E
114°W - 126°W	32	113	12.58°E	13.25°E



Figure 1. Continental U.S. magnetic variation.

ORIGINAL PAGE IS
OF POOR QUALITY



Figure 2. Magnetic variation in the United States from world declination chart (source Defense Mapping Agency).



Figure 3. Block diagram for the microcomputer system.



Figure 4. Logic flow diagrams illustrating steps control program executes to communicate with 9511.



Figure 4. Concluded.

Appendix A. Co-efficients For the 5-band and Secular Change in COT48.

The coefficients (a_{ij}) for the magnetic variation in the conterminous United States (5 bands).

	Band 1	Band 2	Band 3	Band 4	Band 5
a_{00}	-0.12544E 02	-0.26754E 01	0.66671E 01	0.13031E 02	0.15997E 02
a_{10}	0.47404E 00	0.30498E 00	0.73577E-02	-0.24981E 00	-0.40370E 00
a_{11}	-0.79262E 00	-0.85269E 00	-0.62856E 00	-0.36471E 00	-0.20406E 00
a_{20}	-0.14935E-01	-0.11933E-01	-0.75537E-02	0.84073E-03	0.83811E-02
a_{21}	0.47508E-02	0.18570E-01	0.23621E-01	0.19160E-01	0.47466E-02
a_{22}	0.73049E-02	0.19648E-01	-0.99458E-02	-0.17215E-01	-0.65055E-02
a_{30}	0.28976E-03	0.64489E-03	0.57882E-03	0.14505E-03	0.36041E-03
a_{31}	0.46111E-03	-0.73539E-03	-0.29531E-03	-0.10708E-02	-0.16699E-05
a_{32}	-0.15079E-02	-0.16288E-02	0.70147E-03	0.35821E-03	0.17397E-03
a_{33}	0.12362E-02	-0.10931E-02	0.58517E-04	-0.12490E-02	0.22972E-02
a_{40}	0.25381E-04	0.31737E-04	0.54348E-04	0.13665E-04	-0.14418E-04
a_{41}	-0.29623E-04	0.46667E-04	0.10017E-04	-0.63934E-05	-0.40691E-04
a_{42}	-0.18112E-04	-0.51048E-04	0.77275E-04	0.10816E-03	0.88690E-04
a_{43}	0.55077E-04	-0.72203E-04	0.14306E-03	-0.50373E-04	0.18677E-04
a_{44}	0.62795E-04	-0.60910E-03	0.55063E-04	0.31817E-04	-0.51783E-04
a_{50}	0.23134E-06	-0.19467E-05	-0.40952E-05	-0.82114E-06	-0.32150E-05
a_{51}	-0.15270E-05	0.26542E-05	-0.25156E-05	0.60160E-05	0.21843E-05
a_{52}	0.14304E-05	-0.59918E-05	-0.88037E-05	-0.89843E-05	-0.16671E-06
a_{53}	-0.38852E-05	-0.58384E-05	-0.82214E-05	0.65151E-05	-0.13895E-04
a_{54}	0.55360E-05	0.38753E-04	-0.14001E-04	0.58428E-05	0.16509E-04
a_{55}	-0.76365E-05	0.44242E-04	-0.45236E-05	0.19349E-04	-0.37717E-04
a_{60}	-0.62672E-07	-0.76459E-07	-0.12253E-06	-0.20612E-07	0.25454E-07
a_{61}	0.79694E-07	-0.62486E-07	0.17499E-07	-0.41744E-09	0.13351E-06
a_{62}	0.15540E-06	0.17782E-06	0.39251E-07	-0.10090E-06	-0.44300E-06
a_{63}	0.97851E-07	-0.30012E-07	-0.27612E-07	0.57044E-07	0.11357E-05
a_{64}	0.11556E-06	0.24909E-06	-0.15699E-05	-0.84960E-06	-0.14101E-05
a_{65}	-0.76126E-06	0.74355E-06	-0.18538E-05	0.62532E-06	-0.66000E-06
a_{66}	0.23144E-07	0.48778E-05	-0.38992E-06	-0.45290E-07	0.10900E-05
a_{70}	-0.16860E-09	0.24723E-08	0.84712E-08	0.10184E-08	0.65007E-08
a_{71}	0.43030E-09	-0.38370E-08	0.44108E-08	-0.11497E-07	-0.76103E-08
a_{72}	-0.69117E-08	0.19785E-07	0.77909E-08	0.16786E-07	-0.13459E-07
a_{73}	0.33027E-08	-0.76731E-08	-0.37480E-09	0.64112E-09	0.84721E-07
a_{74}	0.93928E-08	-0.40025E-07	0.88535E-07	0.27692E-07	-0.69438E-07
a_{75}	0.91628E-08	0.52668E-07	0.12561E-06	-0.93332E-07	0.29615E-07
a_{76}	0.22806E-07	-0.21183E-06	0.14212E-06	0.52297E-08	-0.85699E-07
a_{77}	0.56746E-08	-0.29999E-06	-0.15083E-07	-0.54997E-07	0.19369E-06

The coefficients (a_{ij}) for the secular change in the cot48 region.

a_{00}	-0.95533E 01
a_{10}	0.11582E 00
a_{11}	-0.93474E-01
a_{20}	0.13750E-01
a_{21}	-0.22416E-01
a_{22}	0.12437E-01
a_{30}	0.27558E-05
a_{31}	0.53560E-03
a_{32}	-0.70816E-03
a_{33}	0.32333E-03
a_{40}	-0.49965E-04
a_{41}	0.18579E-04
a_{42}	0.38544E-05
a_{43}	0.11451E-04
a_{44}	-0.63005E-05
a_{50}	-0.23992E-06
a_{51}	0.35195E-06
a_{52}	-0.26724E-06
a_{53}	-0.53815E-06
a_{54}	0.44979E-06
a_{55}	-0.12145E-06
a_{60}	0.85465E-07
a_{61}	-0.70500E-07
a_{62}	0.26013E-07
a_{63}	0.30705E-09
a_{64}	-0.12121E-07
a_{65}	-0.15529E-08
a_{66}	0.23462E-08

Appendix B. FORTRAN Program Listing of "MAGVAR".

```
C
C     THIS PROGRAM COMPUTES THE MAGNETIC VARIATION AS A
C     FUNCTION OF LATITUDE AND LONGITUDE.  THE SECULAR
C     CHANGE IS ALSO CALCULATED.- IT IS BASED ON THE
C     USD 80 POLYNOMIAL MODEL DEVELOPED BY FABIANO AND
C     OTHERS AT THE UNITED STATES GEOLOGICAL SURVEY IN
C     DENVER CO. PLEASE CONSULT USGS CIRCULAR NO.810 FOR DETAILS.
C                                     RAJAN KAUL - 3/84
C
C     THE INPUT VARIABLES ARE ALAT,ALON AND YEAR
C     REPRESENTING LATITUDE, LONGITUDE AND YEAR.
C     VARIABLES A AND A1 ARE THE COEFFICIENTS TO BE READ
C
      DIMENSION A(8,8),A1(8,8)
      DATA EAST/'EAST'/,WEST/'WEST'/
C
C     READ LATITUDE AND LONGITUDE
C
      WRITE(6,9)
9     FORMAT(1X,'TYPE LAT. AND LONG. AS NNN.NN NNN.NN (F6.2,1X,F6.2)')
      READ(7,8) ALAT,ALON
8     FORMAT(F6.2,1X,F6.2)
C
C     DETERMINE WHICH BAND THE POINT IS IN TO LOAD CORRECT
C     SET OF COEFFICIENTS CORRESPONDING TO PARTICULAR BAND.
C
      IF(ALON.GE.66.0.AND.ALON.LT.78.0) K=11
      IF(ALON.GE.78.0.AND.ALON.LT.90.0) K=12
      IF(ALON.GE.90.0.AND.ALON.LT.102.0) K=13
      IF(ALON.GE.102.0.AND.ALON.LT.114.0) K=14
      IF(ALON.GE.114.0.AND.ALON.LT.126.0) K=15
C
C     READ NORMALIZED LONGITUDE FOR PARTICULAR BAND AND THE
C     COEFFICIENTS.
C
      READ(K,7) DLON
7     FORMAT(F6.2)
      DO 5 NN=1,8
      DO 6 II=1,NN
      READ(K,3,END=13) A(NN,II)
3     FORMAT(E12.5)
6     CONTINUE
5     CONTINUE
C
C     DEFINE COLATITUDE AND NORMALIZED EAST LONGITUDE
C
13    DLA=90.0-ALAT
      DLO=360.0-ALON
C
C     INITIALIZE MAGNETIC VARIATION AND PERFORM CALCULATION
C
      AK=0.0
      DO 1 N=1,8
      DO 2 I=1,N
      KK=IABS(N-I)
      JJ=IABS(I-1)
      DL=DLO-DLON
      IF(DL.EQ.0.0) DL=360.0
      AK=AK+(A(N,I)*((DLA-52.01)**KK)*((DL)**JJ))
2     CONTINUE
1     CONTINUE
C
C     READ COEFFICIENTS FOR SECULAR CHANGE CALCULATION
C
      DO 15 NN=1,7
      DO 16 II=1,NN
      READ(16,23,END=14) A1(NN,II)
23    FORMAT(E12.5)
16    CONTINUE
15    CONTINUE
C
```

```
C
C     DEFINE COLATITUDE AND NORMALIZED EAST LONGITUDE
C
14    DLA=90.0-ALAT
      DLO=360.0-ALON
C
C     INITIALIZE SECULAR CHANGE AND PERFORM CALCULATION
C
      SV=0.0
      DO 11 N=1,7
      DO 12 I=1,N
      KK=IABS(N-I)
      JJ=IABS(I-1)
      DL=DLO-DLON
      IF(DL.EQ.0.0) DL=360.0
      SV=SV+(A(N,I)*((DLA-52.01)**KK)*((DL)**JJ))
12    CONTINUE
11    CONTINUE
C
C     READ YEAR
C
      WRITE(6,17)
17    FORMAT(5X,'TYPE YEAR AS NN.N (F4.1) E.G. JUN 84 = 84.5')
      READ(7,18) YEAR
18    FORMAT(F4.1)
C
C     COMPUTE SECULAR VARIATION ANNUAL AND TO PRESENT DATE.
C     ALSO COMPUTE MAGNETIC VARIATION
C
      SECVAR=SV*(YEAR-85.0)/60.0
      SS=SV/60.0
      VAR=AK+SECVAR
      IF(VAR.LT.0.0) DIR=WEST
      IF(VAR.GT.0.0) DIR=EAST
      V=ABS(VAR)
      WRITE(6,4) ALAT,ALON,V,DIR,SS
4     FORMAT(5X,'LATITUDE = ',F6.2/,5X,'LONGITUDE = ',F6.2/,5X,'MAGNETIC
     & VARIATION = ',F6.2,1X,A4/,5X,'SECULAR CHANGE (ANNUAL) = ',F6.2)
      STOP
      END
```

Appendix C. Instruction Set for the AM9511*.

ORIGINAL PAGE IS OF POOR QUALITY

COMMAND SUMMARY

Command Code 7	6	5	4	3	2	1	0	Command Mnemonic	Command Description
									FIXED-POINT 16-BIT
sr	1	1	0	1	1	0	0	SADD	Add TOS to NOS. Result to NOS. Pop Stack.
sr	1	1	0	1	1	0	1	SSUB	Subtract TOS from NOS. Result to NOS. Pop Stack.
sr	1	1	0	1	1	1	0	SMUL	Multiply NOS by TOS. Lower half of result to NOS. Pop Stack.
sr	1	1	1	0	1	1	0	SMUU	Multiply NOS by TOS. Upper half of result to NOS. Pop Stack.
sr	1	1	0	1	1	1	1	SDIV	Divide NOS by TOS. Result to NOS. Pop Stack.
									FIXED-POINT 32-BIT
sr	0	1	0	1	1	0	0	DADD	Add TOS to NOS. Result to NOS. Pop Stack.
sr	0	1	0	1	1	0	1	DSUB	Subtract TOS from NOS. Result to NOS. Pop Stack.
sr	0	1	0	1	1	1	0	DMUL	Multiply NOS by TOS. Lower half of result to NOS. Pop Stack.
sr	0	1	1	0	1	1	0	DMUU	Multiply NOS by TOS. Upper half of result to NOS. Pop Stack.
sr	0	1	0	1	1	1	1	DDIV	Divide NOS by TOS. Result to NOS. Pop Stack.
									FLOATING-POINT 32-BIT
sr	0	0	1	0	0	0	0	FADD	Add TOS to NOS. Result to NOS. Pop Stack.
sr	0	0	1	0	0	0	1	FSUB	Subtract TOS from NOS. Result to NOS. Pop Stack.
sr	0	0	1	0	0	1	0	FMUL	Multiply NOS by TOS. Result to NOS. Pop Stack.
sr	0	0	1	0	0	1	1	FDIV	Divide NOS by TOS. Result to NOS. Pop Stack.
									DERIVED FLOATING-POINT FUNCTIONS
sr	0	0	0	0	0	0	1	SQRT	Square Root of TOS. Result in TOS.
sr	0	0	0	0	0	1	0	SIN	Sine of TOS. Result in TOS.
sr	0	0	0	0	0	1	1	COS	Cosine of TOS. Result in TOS.
sr	0	0	0	0	1	0	0	TAN	Tangent of TOS. Result in TOS.
sr	0	0	0	0	1	0	1	ASIN	Inverse Sine of TOS. Result in TOS.
sr	0	0	0	0	1	1	0	ACOS	Inverse Cosine of TOS. Result in TOS.
sr	0	0	0	0	1	1	1	ATAN	Inverse Tangent of TOS. Result in TOS.
sr	0	0	0	1	0	0	0	LOG	Common Logarithm (base 10) of TOS. Result in TOS.
sr	0	0	0	1	0	0	1	LN	Natural Logarithm (base e) of TOS. Result in TOS.
sr	0	0	0	1	0	1	0	EXP	Exponential (e^x) of TOS. Result in TOS.
sr	0	0	0	1	0	1	1	PWR	NOS raised to the power in TOS. Result in NOS. Pop Stack.
									DATA MANIPULATION COMMANDS
sr	0	0	0	0	0	0	0	NOP	No Operation
sr	0	0	1	1	1	1	1	FIXS	Convert TOS from floating point to 16-bit fixed point format.
sr	0	0	1	1	1	1	0	FIXD	Convert TOS from floating point to 32-bit fixed point format.
sr	0	0	1	1	1	0	1	FLTS	Convert TOS from 16-bit fixed point to floating point format.
sr	0	0	1	1	1	0	0	FLTD	Convert TOS from 32-bit fixed point to floating point format.
sr	1	1	1	0	1	0	0	CHSS	Change sign of 16-bit fixed point operand on TOS.
sr	0	1	1	0	1	0	0	CHSD	Change sign of 32-bit fixed point operand on TOS.
sr	0	0	1	0	1	0	1	CHSF	Change sign of floating point operand on TOS.
sr	1	1	1	0	1	1	1	PTOS	Push 16-bit fixed point operand on TOS to NOS (Copy)
sr	0	1	1	0	1	1	1	PTOD	Push 32-bit fixed point operand on TOS to NOS. (Copy)
sr	0	0	1	0	1	1	1	PTOF	Push floating point operand on TOS to NOS. (Copy)
sr	1	1	1	1	0	0	0	POPS	Pop 16-bit fixed point operand from TOS. NOS becomes TOS.
sr	0	1	1	1	0	0	0	POPD	Pop 32-bit fixed point operand from TOS. NOS becomes TOS.
sr	0	0	1	1	0	0	0	POPF	Pop floating point operand from TOS. NOS becomes TOS.
sr	1	1	1	1	0	0	1	XCHS	Exchange 16-bit fixed point operands TOS and NOS.
sr	0	1	1	1	0	0	1	XCHD	Exchange 32-bit fixed point operands TOS and NOS.
sr	0	0	1	1	0	0	1	XCHF	Exchange floating point operands TOS and NOS.
sr	0	0	1	1	0	1	0	PUPI	Push floating point constant "π" onto TOS. Previous TOS becomes NOS.

NOTES

1. TOS means Top of Stack. NOS means Next on Stack.
2. AMD Application Brief "Algorithm Details for the Am9511A APU" provides detailed descriptions of each command function, including data ranges, accuracies, stack configurations, etc.
3. Many commands destroy one stack location (bottom of stack) during development of the result. The derived functions may destroy several stack locations. See Application Brief for details.
4. The trigonometric functions handle angles in radians, not degrees.
5. No remainder is available for the fixed point divide functions.
6. Results will be undefined for any combination of command coding bits not specified in this table.

*Original not available at time of publication.

Appendix D. 6502 Assembly Language Program Listing of the Magnetic Variation Implementation on the Ohio University Loran-C Receiver.

```
        ORG $A8
BASE    BSS 2           BASE ADDRESS OF SCRATCHPAD RAM
        ORG $55
CTRN    BSS 1           COUNTER FOR OUTER LOOP IN LEAST SQUARES ALGOTITH
CTRI    BSS 1           COUTNER FOR INNER LOOP IN LEAST SQUARES ALGORITH
COFCTR BSS 1            COUNTER TO POINT AT THE RIGHT COEFFICIENT TO BE
*                       USED IN LEAST SQUARES ALGORITHM
COFTAB BSS 2            ADDRESS OF COEFFICIENT TABLE
MTEMP  BSS 1            TEMPORARY LOCATION USED BY MAGVAR CALCULATION.
MTEMP1 BSS 1            USED BY MAGVAR
*
*   EQUATES TO SUBROUTINE CALLS
*
PUSH    EQU $28AC       SUBROUTINE TO PUSH NUMBER ON TO 9511 STACK
POP     EQU $28DC       SUBROUTINE TO POP NUMBER FROM 9511 STACK
CMND    EQU $290C       SUBROUTINE TO ISSUE COMMAND TO 9511 TO PERFORM
*                      OPERATION
*
*   EQUATES TO VARIABLE ADDRESSES USED IN RNAV
*
PHGS    EQU $E0         LATITUDE OF RECEIVER
THGS    EQU $DC         LONGITUDE OF RECEIVER
P18     EQU $2C         180.0/PI
PAI2    EQU $30         2*PI
F90     EQU $24         PI/2
*          AM9511A COMMANDS.
*
PWR     EQU $0B
FADD    EQU $10
FSUB    EQU $11
FMUL    EQU $12
FDIV    EQU $13
SQRT    EQU 1
CHSF    EQU $15
*          CONSTANTS AND VARIABLES
BAND    EQU $0              ADDR FOR NORMALIZED LONGITUDE FOR PARTICULAR BAN
*            FOLLOWED BY 36 CO-EFFICIENTS FOR EACH BAND AT LOCATION
*            $C800 TO $CCFF - ONE PAGE FOR EACH OF 5 BANDS.
*
*
*   CONSTANTS FOR DIVISION IN LEAST SQUARES ALGORITHM FOR MAGVAR.
*
MZERO   EQU $0          -0.0
MONE    EQU MZERO+4     - 1.0
MTWO    EQU MONE+4      -2.0
MTHREE EQU MTWO+4       -3.0
MFOUR   EQU MTHREE+4    -4.0
MFIVE   EQU MFOUR+4     -5.0
MSIX    EQU MFIVE+4     -6.0
MSEVEN EQU MSIX+4       -7.0
*
*   CONSTANTS THAT DEFINE LIMITS IN BANDS OF COT48 TO DETERMINE
*   WHICH SET OF CO-EFFICIENTS NEED TO BE USED IN THE ALGORITHM
*   TO DETERMINE MAGNETIC VARIATION.
*
A78     EQU MSEVEN+4
A90     EQU A78+4
A102    EQU A90+4
A114    EQU A102+4
A5201   EQU A114+4      NORMALIZED LATITUDE = 52.01*PI/180 RADIANS
F180    EQU A5201+4     PI/180
ATEMP   EQU F180+4      TEMPORARY LOCATION USED WHILE DETERMINING THE PA
NDL     EQU ATEMP+4     MDLO-NORMALIZED LONGITUDE FOR PARTICULAR BAND.
MDLA    EQU NDL+4       PI/2-PHGS (DEGREES)
MDLO    EQU MDLA+4      2*PI-THGS (DEGREES)
MDLA52 EQU MDLO+4       MDLA-A5201
MAGVAR EQU MDLA52+4     CUMULATIVE MAGNETIC VARIATION
RLTEMP EQU MAGVAR+4     TEMPERORY LOCATION
MAGVD   EQU RLTEMP+4    MAGNETIC VARIATION
*
*
*
        ORG $C000
```

```
*
*   MOVE CONSTANT NUMBER TABLE IN SCRATCH SPACE
*
        LDA =1
        STA BASE+1      BASE = $0100
        LDA =0
        LDY =0
MO      LDA TABLE1,Y
        STA (BASE),Y
        INY
        CPY =56
        BNE MO
*
*
*   MAGNETIC VARIATION CALCULATION
*
        LDA =0
        STA CTRN
        STA CTRI        INITIALIZE COUNTERS
*
*   CALCULATE MDLA
*
        INC BASE+1
        INC BASE+1      BASE = $0300
        LDY =F90
        JSR PUSH
        DEC BASE+1      BASE = $0200
        LDY =PHGS
        JSR PUSH
        LDA =FSUB
        JSR CMND
        INC BASE+1      BASE = $0300
        LDY =P18
        JSR PUSH
        LDA =FMUL
        JSR CMND
        DEC BASE+1
        DEC BASE+1      BASE = $0100
        LDY =MDLA       MDLA = 90-PHGS
        JSR POP
*
*   CALCULATE MDLO
*
        INC BASE+1
        INC BASE+1      BASE = $0300
        LDY =PAI2
        JSR PUSH
        DEC BASE+1      BASE = $0200
        LDY =THGS
        JSR PUSH
        LDA =FSUB
        JSR CMND
        INC BASE+1      BASE = $0300
        LDY =P18
        JSR PUSH
        LDA =FMUL
        JSR CMND
        DEC BASE+1
        DEC BASE+1      BASE = $0100
        LDY =MDLO
        JSR POP         MDLO = 360-THGS
*
*   CALCULATE MDLA52
*
        LDY =MDLA
        JSR PUSH
        LDY =A5201
        JSR PUSH
        LDA =FSUB
        JSR CMND
        JSR CMND
        LDY =MDLA52
        JSR POP         MDLA52 = MDLA-52
```

```
*
        INC BASE+1       BASE = $0200
        LDY =THGS
        JSR PUSH
        DEC BASE+1       BASE = $0100
        LDY =RLTEMP
        JSR POP          PUT THGS IN RLTEMP FOR COMPARISION PURPOSES IN T
*                   HE NEXT SEGMENT TO DETERMINE WHICH BAND TO USE
*              TO CALCULATE MAGVAR.
*
*   DETERMINE WHICH BAND IT IS TO CALCULATE MAGVAR.
*
        LDY =A78
        JSR PUSH
        LDY =RLTEMP
        JSR PUSH
        LDA =FSUB
        JSR CMND
        LDY =ATEMP
        JSR POP          ATEMP = 78*PI/180 - THGS
        LDY =ATEMP
        LDA (BASE),Y
        BPL M1           IF ATEMP IS +VE -- BAND 1, IF NOT TRY FOR BAND 2
*
        LDY =A90
        JSR PUSH
        LDY =RLTEMP
        JSR PUSH
        LDA =FSUB
        JSR CMND
        LDY =ATEMP
        JSR POP          ATEMP = 90*PI/180 - THGS
        LDY =ATEMP
        LDA (BASE),Y
        BPL M2           IF ATEMP IS +VE - BAND 2, IF NOT TRY FOR BAND 3
*
        LDY =A102
        JSR PUSH
        LDY =RLTEMP
        JSR PUSH
        LDA =FSUB
        JSR CMND
        LDY =ATEMP
        JSR POP          ATEMP = 102*PI/180 - THGS
        LDY =ATEMP
        LDA (BASE),Y
        BPL M3           IF ATEMP IS +VE - BAND 3, IF NOT TRY BAND 4
*
        LDY =A114
        JSR PUSH
        LDY =RLTEMP
        JSR PUSH
        LDA =FSUB
        JSR CMND
        LDY =ATEMP
        JSR POP          ATEMP = 114*PI/180 - THGS
        LDY =ATEMP
        LDA (BASE),Y
        BPL M4           IF ATEMP IS +VE - BAND 4
        JMP M5           MUST BE BAND 5
*
*   SET CO-EFFICIENT TABLE ADDRESS TO CORRESPOND WITH PARTICULAR BAND
*
M1      LDY =MDLO
        JSR PUSH
        LDA =$C8
        STA BASE+1
        STA COFTAB+1     BAND 1
        JMP M6
*
M2      LDY =MDLO
        JSR PUSH
        LDA =$C9
```

```
        STA BASE+1
        STA COFTAB+1     BAND 2
        JMP M6
*
M3      LDY =MDLO
        JSR PUSH
        LDA =$CA
        STA BASE+1
        STA COFTAB+1     BAND 3
        JMP M6
*
M4      LDY =MDLO
        JSR PUSH
        LDA =$CB
        STA BASE+1
        STA COFTAB+1     BAND 4
        JMP M6
*
M5      LDY =MDLO
        JSR PUSH
        LDA =$CC
        STA BASE+1
        STA COFTAB+1     BAND 5
*
M6      LDA =0
        STA COFTAB
        LDY =BAND
        JSR PUSH
        LDA =FSUB
        JSR CMND
        LDA =1
        STA BASE+1       BASE = $0100
        LDY =NDL
        JSR POP          NDL=MDLO-NORMALIZED LONGITUDE FOR PARTICULAR BAN
        LDA =4
        STA COFCTR       SET CO-EFFICIENT COUNTER TO POINT TO CO-EFFICIEN
C2      CLC
        LDA CTRN         LOAD OUTER LOOP COUNTER
        ROL A
        ROL A            POINT TO LOCATION FOR EXPONENTS FOR LEAST SQUARE
        STA MTEMP
        LDY MTEMP
        JSR PUSH
        LDA CTRI         LOAD INNER LOOP COUNTER
        ROL A
        ROL A            POINT TO LOCATION FOR EXPONENTS
        STA MTEMP
        LDY MTEMP
        JSR PUSH
        LDA =FSUB
        JSR CMND         N-1
        LDY =RLTEMP
        JSR POP          (N-1)
        LDY =MDLA52
        JSR PUSH
        LDY =MDLA52
        LDA (BASE),Y
        BPL C6
        LDA =1           SET FLAG IF NEGATIVE AND CHANGE SIGN
        STA MTEMP1
        LDA =CHSF
        JSR CMND
        JMP C9
C6      LDA =0
        STA MTEMP1
C9      LDY =RLTEMP
        JSR PUSH
        LDA =PWR
        JSR CMND         MDLA52**(N-1)
        CLC
        LDA CTRN
        SBC CTRI
        AND =1           EXPONENT EVEN ?
```

```
        BNE C4          YES, LOOP OUT
        LDA MTEMP1      NO, IS NEGATIVE FLAG SET ?
        BEQ C4          NO, FLAG NOT SET -- LOOP OUT
        LDA =CHSF       EXPONENT WAS ODD AND NEGATIVE FLAG
        JSR CMND        WAS SET --- THEREFORE CHANGE SIGN AGAIN
C4      LDY =ATEMP
        JSR POP         MDLA52**(N-I)
        CLC
        LDA CTRI        LOAD INNER LOOP COUNTER FOR LEAST SQUARES PROCED
        ROL A
        ROL A           POINT TO LOCATION FOR EXPONENTS
        STA MTEMP
        LDY MTEMP
        JSR PUSH
        LDY =RLTEMP
        JSR POP
        LDY =NDL
        JSR PUSH
        LDY =NDL
        LDA (BASE),Y
        BPL C7
        LDA =0
        STA MTEMP1
        LDA =CHSF
        JSR CMND
        JMP C8
C7      LDA =1
        STA MTEMP1
C8      LDY =RLTEMP
        JSR PUSH
        LDA =PWR
        JSR CMND        NDL**I
*
* IN THIS NEXT SEGMENT A TEST IS DONE TO MAKE SURE THE CORRECT SIGN
* IS ATTACHED WITH THE RESULT AFTER THE EXPONENT CALCULATION.
*
        LDA CTRI
        AND =1
        BEQ C5
        LDA MTEMP1
        BNE C5
        LDA =CHSF
        JSR CMND
C5      LDY =ATEMP
        JSR PUSH
        LDA =FMUL
        JSR CMND        MDLA52**(N-I)*NDL**I
        LDA COFTAB+1
        STA BASE+1      PUT CO-EFFICIENT TABLE ADDRESS IN BASE
        LDY COFCTR      POINT TO CO-EFFICIENT COUNTER
        JSR PUSH
        LDA =FMUL
        JSR CMND        A(NI)*NDL**(I)*MDLA52**(N-I)
        LDA =1
        STA BASE+1      BASE = $0100
        LDY =MAGVAR
        JSR PUSH
        LDA =FADD
        JSR CMND        ADD TO ACCUMULATE THE VALUE OF MAGVAR
        LDY =MAGVAR
        JSR POP         MAGVAR=MAGVAR+A(N,I)*NDL**I*MDLA52**(N-I)
        INC COFCTR
        INC COFCTR
        INC COFCTR
        INC COFCTR      POINT TO NEXT SET OF CO-EFFICIENTS
        LDA CTRN
        CMP CTRI
        BEQ C1          IF THEY ARE EQUAL INNER LOOP DONE, CHECK IF OUTE
        INC CTRI
        JMP C2          IF NOT, GO BACK AND COMPLETE OUTER LOOP
C1      LDA CTRN        CHECK TO SEE IF OUTER LOOP COMPLETE
        CMP =7
        BEQ C3          OUTER LOOP ALSO DONE, BRANCH OUT
```

```
        INC CTRN          INCREMENT OUTER LOOP COUNTER
        LDA =0
        STA CTRI          INITIALIZE INNER LOOP COUNTER
        JMP C2            START OVER
C3      RTS               RETURN FROM MAGVAR TO MAIN PROGRAM
*
*    TABLE OF CONSTANTS USED BY MAGNETIC VARIATION STORED
*    STARTING AT $0100 IN SCRATCHPAD RAM LOCATION.
*
TABLE1 HEX 00,00,00,00 - 0.0 (MZERO)
        HEX 01,80,00,00 - 1.0 (MONE)
        HEX 02,80,00,00 - 2.0 (MTWO)
        HEX 02,C0,00,00 - 3.0 (MTHREE)
        HEX 03,80,00,00 - 4.0 (MFOUR)
        HEX 03,A0,00,00 - 5.0 (MFIVE)
        HEX 03,C0,00,00 - 6.0 (MSIX)
        HEX 03,E0,00,00 - 7.0 (MSEVEN)
        HEX 01,AE,40,F1 - 78*PI/180 (A78)
        HEX 01,C9,0F,DA - 90*PI/180 (A90)
        HEX 01,E3,DE,C4 - 102*PI/180 (A102)
        HEX 01,FE,AD,AE - 114*PI/180 (A114)
        HEX 06,D0,0A,3D - 52.01*PI/180 (A5201)
        HEX 7B,8E,FA,35 - PI/180.0 (F180)
*
*    CO-EFFICIENTS FOR THE FIVE BANDS OF THE COT48
*    THE DATA LABELED BAND(N) IS THE NORMALIZED
*    LONGITUDE FOR EACH BAND. THE CO-EFFICIENTS
*    ARE STORED IN ONE PAGE CHUNKS STARTING AT
*    $3800 TO $3CFF.
*
        ORG $C800
BAND1   HEX 09,90,81,48 - 289.01
        HEX 84,C8,B2,96
        HEX 7F,F2,B6,07
        HEX 80,CA,E9,68
        HEX FA,F4,B3,9C
        HEX 79,9B,AC,C4
        HEX 79,EF,5E,07
        HEX 75,97,EB,10
        HEX 75,F1,C0,9A
        HEX F7,C5,A5,66
        HEX 77,A2,06,A6
        HEX 71,D4,E9,7F
        HEX F1,F8,7E,E8
        HEX F1,97,F0,15
        HEX 72,E7,02,28
        HEX 73,83,B0,EB
        HEX 6A,F8,66,41
        HEX ED,CC,F2,4A
        HEX 6D,BF,FC,83
        HEX EF,82,5D,9D
        HEX 6F,B9,C1,F7
        HEX F0,80,1E,7A
        HEX E9,86,96,3E
        HEX 69,AB,24,13
        HEX 6A,A6,DD,20
        HEX 69,D2,22,6E
        HEX 69,F8,29,C8
        HEX EC,CC,59,1D
        HEX 67,C6,CD,89
        HEX E0,B9,62,17
        HEX 61,EC,8F,0F
        HEX E5,ED,7C,38
        HEX 64,E2,F5,2F
        HEX 66,A1,5E,31
        HEX 66,9D,6A,60
        HEX 67,C3,E7,CE
        HEX 65,C2,FA,4F
        ORG $C900
BAND2   HEX 09,8A,81,48 - 277.01
        HEX 82,AB,39,97
        HEX 7F,9C,26,DD
        HEX 80,DA,4A,06
```

```
         HEX FA,C3,82,A1
         HEX 7B,98,1F,46
         HEX 7B,A0,F5,0D
         HEX 76,A9,0D,F8
         HEX F6,C0,C6,EB
         HEX F7,D5,7C,74
         HEX F7,8F,48,07
         HEX 72,85,1D,21
         HEX 72,C3,BC,D9
         HEX F2,D6,1C,5F
         HEX F3,97,6B,BF
         HEX F6,9F,AC,25
         HEX EE,82,A3,89
         HEX 6E,B2,1F,27
         HEX EF,C9,0D,58
         HEX EF,C3,E7,79
         HEX 72,A2,8A,81
         HEX 72,B9,90,7C
         HEX E9,A4,31,BA
         HEX E9,86,30,51
         HEX 6A,BE,EF,E4
         HEX E8,80,E6,8B
         HEX 6B,85,BA,65
         HEX 6C,C7,98,81
         HEX 6F,A3,AC,54
         HEX 64,A9,E4,CF
         HEX E5,83,D6,58
         HEX 67,A9,F4,1D
         HEX E6,83,D2,DF
         HEX E8,AB,E8,48
         HEX 68,E2,35,A1
         HEX EA,E3,72,90
         HEX EB,A1,0D,C3
         ORG $CA00
BAND3    HEX 09,84,81,48 - 265.01
         HEX 03,D5,58,CD
         HEX 79,F1,19,1D
         HEX 80,A0,E9,70
         HEX F9,F7,84,B1
         HEX 7B,C1,81,09
         HEX FA,A2,F3,EB
         HEX 76,97,BB,AF
         HEX F5,9A,D3,6F
         HEX 76,B7,E2,A7
         HEX 72,F5,70,94
         HEX 72,E3,F3,4B
         HEX 70,A8,0F,1E
         HEX 73,A2,0E,7C
         HEX 74,96,03,2F
         HEX 72,E6,F3,C5
         HEX EF,89,69,28
         HEX EE,A8,D1,DA
         HEX F0,93,B3,7A
         HEX F0,89,EE,A5
         HEX F0,EA,E5,68
         HEX EF,97,C9,C6
         HEX EA,83,91,11
         HEX 67,96,50,EE
         HEX 68,A8,95,5F
         HEX E7,ED,2E,8D
         HEX ED,D2,B5,5A
         HEX ED,F8,D0,15
         HEX EB,D1,56,79
         HEX 66,91,88,D2
         HEX 65,97,8D,FC
         HEX 66,85,D8,91
         HEX E1,CE,0C,45
         HEX 69,BE,20,A2
         HEX 6A,86,DF,B0
         HEX 6A,98,99,19
         HEX E7,81,8F,35
         ORG $CB00
```

```
BAND4   HEX 08,FD,02,8F - 253.01
        HEX 04,D0,80,9D
        HEX FE,FF,CE,74
        HEX FF,BA,BB,CB
        HEX 76,DC,64,B4
        HEX 7B,9C,F5,DA
        HEX FB,8D,07,85
        HEX 74,98,19,1A
        HEX F7,8C,5B,C3
        HEX 75,BB,CE,22
        HEX F7,A3,B4,25
        HEX 70,E5,42,BA
        HEX EF,D6,86,8B
        HEX 73,E2,D2,E2
        HEX F2,D3,47,EB
        HEX 72,85,73,91
        HEX EC,DC,6C,A8
        HEX 6F,C9,DD,4E
        HEX F0,96,BB,67
        HEX 6F,DA,9C,74
        HEX 6F,C4,0D,19
        HEX 71,A2,4F,72
        HEX E7,B1,0E,7D
        HEX E1,E5,7D,2D
        HEX E9,D8,AD,43
        HEX 68,F5,00,52
        HEX EC,E4,10,69
        HEX 6C,A7,DC,02
        HEX E8,C2,84,C6
        HEX 63,8B,F8,78
        HEX E6,C5,86,8A
        HEX 67,90,31,E7
        HEX 62,B0,3B,29
        HEX 67,ED,DE,B1
        HEX E9,C8,6D,A5
        HEX 65,B3,B1,5E
        HEX E8,EC,36,48
        ORG $CC00
BAND5   HEX 08,F1,02,8F - 241.01
        HEX 04,FF,F4,F1
        HEX FF,CE,B2,29
        HEX FE,D0,F4,95
        HEX 7A,89,50,CC
        HEX 79,9B,89,C9
        HEX F9,D5,2C,6D
        HEX 75,BC,F5,25
        HEX ED,E0,20,A4
        HEX 74,B6,6B,73
        HEX 78,96,8C,74
        HEX F0,F1,E6,21
        HEX F2,AA,AC,0F
        HEX 73,B9,FF,23
        HEX 71,9C,AD,69
        HEX F2,D9,31,E5
        HEX EE,D7,C0,F0
        HEX 6E,92,95,4D
        HEX EA,B3,02,45
        HEX F0,E9,1C,DA
        HEX 71,8A,7C,2A
        HEX F2,9E,31,E4
        HEX ED,D2,B5,5A
        HEX ED,F8,D0,15
        HEX EB,D1,56,79
        HEX 66,91,88,D2
        HEX 65,97,8D,FC
        HEX 66,85,D8,91
        HEX E1,CE,0C,45
        HEX 69,BE,20,A2
        HEX 6A,86,DF,B0
        HEX 6A,98,99,19
        HEX E7,81,8F,35
        ORG $CB00
```

```
BAND4   HEX 08,FD,02,8F - 253.01
        HEX 04,D0,80,9D
        HEX FE,FF,CE,74
        HEX FF,BA,BB,CB
        HEX 76,DC,64,B4
        HEX 7B,9C,F5,DA
        HEX FB,8D,07,85
        HEX 74,98,19,1A
        HEX F7,8C,5B,C3
        HEX 75,BB,CE,22
        HEX F7,A3,B4,25
        HEX 70,E5,42,BA
        HEX EF,D6,86,8B
        HEX 73,E2,D2,E2
        HEX F2,D3,47,EB
        HEX 72,85,73,91
        HEX EC,DC,6C,A8
        HEX 6F,C9,DD,4E
        HEX F0,96,BB,67
        HEX 6F,DA,9C,74
        HEX 6F,C4,0D,19
        HEX 71,A2,4F,72
        HEX E7,B1,0E,7D
        HEX E1,E5,7D,2D
        HEX E9,D8,AD,43
        HEX 68,F5,00,52
        HEX EC,E4,10,69
        HEX 6C,A7,DC,02
        HEX E8,C2,84,C6
        HEX 63,8B,F8,78
        HEX E6,C5,86,8A
        HEX 67,90,31,E7
        HEX 62,B0,3B,29
        HEX 67,ED,DE,B1
        HEX E9,C8,6D,A5
        HEX 65,B3,B1,5E
        HEX E8,EC,36,48
        ORG $CC00
BAND5   HEX 08,F1,02,8F - 241.01
        HEX 04,FF,F4,F1
        HEX FF,CE,B2,29
        HEX FE,D0,F4,95
        HEX 7A,89,50,CC
        HEX 79,9B,89,C9
        HEX F9,D5,2C,6D
        HEX 75,BC,F5,25
        HEX ED,E0,20,A4
        HEX 74,B6,6B,73
        HEX 78,96,8C,74
        HEX F0,F1,E6,21
        HEX F2,AA,AC,0F
        HEX 73,B9,FF,23
        HEX 71,9C,AD,69
        HEX F2,D9,31,E5
        HEX EE,D7,C0,F0
        HEX 6E,92,95,4D
        HEX EA,B3,02,45
        HEX F0,E9,1C,DA
        HEX 71,8A,7C,2A
        HEX F2,9E,31,E4
        HEX 67,DA,A5,80
        HEX 6A,8F,5A,F3
        HEX EB,ED,D6,25
        HEX 6D,98,6D,AB
        HEX ED,BD,43,5B
        HEX EC,B1,2A,FE
        HEX 6D,92,4C,1D
        HEX 65,DF,5D,34
        HEX E6,82,BE,74
        HEX E6,E7,39,BF
        HEX 69,B5,F0,07
        HEX E9,95,1D,C7
        HEX 67,FE,63,A2
        HEX E9,B8,09,B0
        HEX 6A,CF,F8,D4
        END
```